S = 1/2 Heisenberg model

$H_{\text{Heisenberg}} = J \sum_{<i,j>} S_i . S_j$

Basis representation : Direct product of spinors for both sites

```
S12basis = {
   {{0, 1}, {0, 1}}, {{0, 1}, {1, 0}}, {{1, 0}, {0, 1}}, {{1, 0}, {1, 0}}};
```

```
HS12[f_, i_] :=
 J/4 Sum[ f[[1]].PauliMatrix[α].i[[1]] * f[[2]].PauliMatrix[α].i[[2]], {α, 3} ]
HS12matrix = Table[ HS12[f, i], {f, S12basis}, {i, S12basis}];
HS12matrix // MatrixForm
```

$$\begin{pmatrix} \frac{J}{4} & 0 & 0 & 0 \\ 0 & -\frac{J}{4} & \frac{J}{2} & 0 \\ 0 & \frac{J}{2} & -\frac{J}{4} & 0 \\ 0 & 0 & 0 & \frac{J}{4} \end{pmatrix}$$

Sectors

The total spin component along one direction is conserved,
hence we can split the Hamiltonian into sectors $S_z^{(tot)} = \{-1, 0, 1\}$.

```
Module[{basis, H, evals, evecs},
 Do[
  Print["Sz=", (basis[[1]][[1]].PauliMatrix[3].basis[[1]][[1]] +
      basis[[1]][[2]].PauliMatrix[3].basis[[1]][[2]]) / 2, ":"];
  Print["\tStates: ", basis];
  H = Table[ HS12[f, i], {f, basis}, {i, basis}];
  Print["\tH:", MatrixForm[H]];
  {evals, evecs} = Eigensystem[H];
  Print["\tEigenstates:"];
  Do[Print["\t\tE=", evals[[i]], ": ", evecs[[i]]], {i, Length[evals]}];
  , {basis, {{S12basis[[1]]}, {S12basis[[2]], S12basis[[3]]}, {S12basis[[4]]}}}
 ]
]
```

$S_z = -1$:

States: {{{0, 1}, {0, 1}}}

H: $\begin{pmatrix} \frac{J}{4} \end{pmatrix}$

Eigenstates:

$E = \frac{J}{4}$: {1}

$S_z = 0$:

States: {{{0, 1}, {1, 0}}, {{1, 0}, {0, 1}}}

H: $\begin{pmatrix} -\frac{J}{4} & \frac{J}{2} \\ \frac{J}{2} & -\frac{J}{4} \end{pmatrix}$

Eigenstates:

$E = -\frac{3J}{4}$: {-1, 1}

$E = \frac{J}{4}$: {1, 1}

$S_z = 1$:

States: {{{1, 0}, {1, 0}}}

H: $\begin{pmatrix} \frac{J}{4} \end{pmatrix}$

Eigenstates:

$E = \frac{J}{4}$: {1}

We see the $S = 1$ triplet with energy $\frac{J}{4}$ and the $S = 0$ singlet at $E = -\frac{3}{4} J$. The former is the ground state for ferromagnetic coupling $(J < 0)$, the latter for antiferromagnetic $(J > 0)$.

S = 1 Heisenberg model

Basis representation :
Direct product of S_z eigenstates $\{ | m >, m = -1, 0, 1\}$ for both sites

```
S1basis = Flatten[Table[{m, n}, {m, -1, 1}, {n, -1, 1}], 1]
```

{{-1, -1}, {-1, 0}, {-1, 1}, {0, -1}, {0, 0}, {0, 1}, {1, -1}, {1, 0}, {1, 1}}

```
Splus[mf_, mi_] := Sqrt[2 - mi (mi + 1)] KroneckerDelta[mf, mi + 1]
Sminus[mf_, mi_] := Sqrt[2 - mi (mi - 1)] KroneckerDelta[mf, mi - 1]
Sx[i__] := (Splus[i] + Sminus[i]) / 2
Sy[i__] := (Splus[i] - Sminus[i]) / (2 I)
Sz[mf_, mi_] := mi KroneckerDelta[mf, mi]
HS1[{mf_, nf_}, {mi_, ni_}] :=
 J (Sx[mf, mi] Sx[nf, ni] + Sy[mf, mi] Sy[nf, ni] + Sz[mf, mi] Sz[nf, ni])
HS1matrix = Table[HS1[f, i], {f, S1basis}, {i, S1basis}];
```

```
HS1matrix // MatrixForm
```

$$\begin{pmatrix} J & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & J & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & -J & 0 & J & 0 & 0 & 0 & 0 \\ 0 & J & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & J & 0 & 0 & 0 & J & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & J & 0 \\ 0 & 0 & 0 & 0 & J & 0 & -J & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & J & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & J \end{pmatrix}$$

Sectors

Again : The total spin component along one direction is conserved, hence we can split the Hamiltonian into sectors $S_z^{(tot)} = \{-2, -1, 0, 1, 2\}$.

```
Module[{sz, basis, H, evals, evecs},
 Do[
  Print["Sz = ", sz, ":"];
  basis = Select[S1basis, #[[1]] + #[[2]] == sz &];
  Print["\tBasis: ", basis];
  H = Table[HS1[f, i], {f, basis}, {i, basis}];
  Print["\tHamiltonian: ", MatrixForm[H]];
  Print["\tEigensystem: ", Eigensystem[H]];
  , {sz, -2, 2}
 ]
]
```

$S_z = -2$:

Basis: {{-1, -1}}

Hamiltonian: $\begin{pmatrix} J \end{pmatrix}$

Eigensystem: {{J}, {{1}}}

$S_z = -1$:

Basis: {{-1, 0}, {0, -1}}

Hamiltonian: $\begin{pmatrix} 0 & J \\ J & 0 \end{pmatrix}$

Eigensystem: {{-J, J}, {{-1, 1}, {1, 1}}}

$S_z = 0$:

Basis: {{-1, 1}, {0, 0}, {1, -1}}

Hamiltonian: $\begin{pmatrix} -J & J & 0 \\ J & 0 & J \\ 0 & J & -J \end{pmatrix}$

Eigensystem: {{-2 J, -J, J}, {{1, -1, 1}, {-1, 0, 1}, {1, 2, 1}}}

$S_z = 1$:

Basis: {{0, 1}, {1, 0}}

Hamiltonian: $\begin{pmatrix} 0 & J \\ J & 0 \end{pmatrix}$

Eigensystem: {{-J, J}, {{-1, 1}, {1, 1}}}

$S_z = 2$:

Basis: {{1, 1}}

Hamiltonian: $\begin{pmatrix} J \end{pmatrix}$

Eigensystem: {{J}, {{1}}}

Bose - Hubbard model

$$H_{BH} = -t \sum_{<i,j>} \left(b_i^\dagger b_j + b_i b_j^\dagger\right) + \frac{U}{2} \sum_i n_i (n_i - 1) - \mu \sum_i n_i$$

■ Basis

The basis can be enumerated by the number of particles on each site.
We employ a maximum number of particles in the system $N_{max} = 4$.

```
maxN = 4; (* Particle number cut-off *)
```

```
BHBasis = Flatten[Table[{k, n - k}, {n, 0, maxN}, {k, 0, n}], 1]
```

```
{{0, 0}, {0, 1}, {1, 0}, {0, 2}, {1, 1}, {2, 0}, {0, 3},
 {1, 2}, {2, 1}, {3, 0}, {0, 4}, {1, 3}, {2, 2}, {3, 1}, {4, 0}}
```

■ Hamiltonian

Creation/annihilation operators

```
Bcrt[f_, i_] := √f KroneckerDelta[f, i + 1];
Bann[f_, i_] := √i KroneckerDelta[f, i - 1];

HBH[{k_, l_}, {m_, n_}] := -t (Bcrt[k, m] Bann[l, n] + Bann[k, m] Bcrt[l, n]) +
   U/2 (m (m - 1) + n (n - 1)) DiscreteDelta[k - m, l - n];
HBHmatrix = Table[HBH[f, i], {f, BHBasis}, {i, BHBasis}];
HBHmatrix // MatrixForm
```

$$\begin{pmatrix}
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & U & -\sqrt{2}\,t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -\sqrt{2}\,t & 0 & -\sqrt{2}\,t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & -\sqrt{2}\,t & U & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 3U & -\sqrt{3}\,t & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -\sqrt{3}\,t & U & -2t & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & -2t & U & -\sqrt{3}\,t & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\sqrt{3}\,t & 3U & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 6U & -2t & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2t & 3U & -\sqrt{6}\,t & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\sqrt{6}\,t & 2U & -\sqrt{6}\,t \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -\sqrt{6}\,t & 3U \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -2t
\end{pmatrix}$$

■ Sectors

The Hamiltonian conserves the total particle number (U (1) symmetry), hence we can split into N - sectors

```
doBHSector[n_] := Module[{basis, hmatrix, evals, evecs},
  Print["N = ", n, ":"];
  basis = Select[BHBasis, #[[1]] + #[[2]] == n &];
  Print["\tBasis: ", basis];
  hmatrix = Table[HBH[f, i], {f, basis}, {i, basis}];
  Print["\tHamiltonian: ", MatrixForm[hmatrix]];
  Do[
   {evals, evecs} = Eigensystem[hmatrix /. {t → 1., U → u}];
   Print["\tU=", u, ":"];
   Do[Print["\t\tE=", evals[[i]], "\t", evecs[[i]]], {i, Length[evals]}]
   , {u, {-1, 1, 4}}
  ]
 ]
```

```
For[n = 0, n ≤ maxN, n++, doBHSector[n]]
```

N = 0:

Basis: {{0, 0}}

Hamiltonian: $\begin{pmatrix} 0 \end{pmatrix}$

U=−1:

E=0 {1}

U=1:

E=0 {1}

U=4:

E=0 {1}

N = 1:

Basis: {{0, 1}, {1, 0}}

Hamiltonian: $\begin{pmatrix} 0 & -t \\ -t & 0 \end{pmatrix}$

U=−1:

E=−1. {−0.707107, −0.707107}

E=1. {−0.707107, 0.707107}

U=1:

E=−1. {−0.707107, −0.707107}

E=1. {−0.707107, 0.707107}

U=4:

E=−1. {−0.707107, −0.707107}

E=1. {−0.707107, 0.707107}

N = 2:

Basis: {{0, 2}, {1, 1}, {2, 0}}

Hamiltonian: $\begin{pmatrix} U & -\sqrt{2}\,t & 0 \\ -\sqrt{2}\,t & 0 & -\sqrt{2}\,t \\ 0 & -\sqrt{2}\,t & U \end{pmatrix}$

U=−1:

E=−2.56155 {0.557345, 0.615412, 0.557345}

E=1.56155 {0.435162, −0.788205, 0.435162}

E=−1. $\{0.707107, -8.88178\times10^{-16}, -0.707107\}$

U=1:

E=2.56155 {−0.557345, 0.615412, −0.557345}

E=−1.56155 {0.435162, 0.788205, 0.435162}

E=1. $\{0.707107, -5.55112\times10^{-16}, -0.707107\}$

U=4:

E=4.82843 {0.653281, −0.382683, 0.653281}

E=4. $\{0.707107, -2.77556\times10^{-16}, -0.707107\}$

E=−0.828427 {0.270598, 0.92388, 0.270598}

N = 3:

Basis: {{0, 3}, {1, 2}, {2, 1}, {3, 0}}

Hamiltonian: $\begin{pmatrix} 3\,U & -\sqrt{3}\,t & 0 & 0 \\ -\sqrt{3}\,t & U & -2\,t & 0 \\ 0 & -2\,t & U & -\sqrt{3}\,t \\ 0 & 0 & -\sqrt{3}\,t & 3\,U \end{pmatrix}$

U=−1:

E=−4.73205 {0.5, 0.5, 0.5, 0.5}

E=−3.64575 {0.662557, 0.247018, −0.247018, −0.662557}

E=1.64575 {0.247018, −0.662557, 0.662557, −0.247018}

E=−1.26795 {0.5, −0.5, −0.5, 0.5}

U=1:

E=4.73205 {0.5, −0.5, 0.5, −0.5}

E=3.64575 {0.662557, −0.247018, −0.247018, 0.662557}

E=−1.64575 {0.247018, 0.662557, 0.662557, 0.247018}

E=1.26795 {−0.5, −0.5, 0.5, 0.5}

U=4:

E=12.4641 {−0.683013, 0.183013, −0.183013, 0.683013}

E=12.2915 {0.6973, −0.117355, −0.117355, 0.6973}

E=5.5359 {0.183013, 0.683013, −0.683013, −0.183013}

E=1.7085 {0.117355, 0.6973, 0.6973, 0.117355}

N = 4:

Basis: {{0, 4}, {1, 3}, {2, 2}, {3, 1}, {4, 0}}

Hamiltonian: $\begin{pmatrix} 6\,U & -2\,t & 0 & 0 & 0 \\ -2\,t & 3\,U & -\sqrt{6}\,t & 0 & 0 \\ 0 & -\sqrt{6}\,t & 2\,U & -\sqrt{6}\,t & 0 \\ 0 & 0 & -\sqrt{6}\,t & 3\,U & -2\,t \\ 0 & 0 & 0 & -2\,t & 6\,U \end{pmatrix}$

U=−1:

E=−7.61463 {0.513009, 0.414159, 0.36137, 0.414159, 0.513009}

E=−7. $\{-0.632456, -0.316228, -2.86658\times10^{-16}, 0.316228, 0.632456\}$

E=−4.63268 {−0.468641, 0.320392, 0.596196, 0.320392, −0.468641}

E=−2. $\{0.316228, -0.632456, 3.4399\times10^{-16}, 0.632456, -0.316228\}$

E=1.2473 {0.13114, −0.475207, 0.716911, −0.475207, 0.13114}

U=1:

E=7.61463 {0.513009, −0.414159, 0.36137, −0.414159, 0.513009}

E=7. $\{-0.632456, 0.316228, -2.86658\times10^{-16}, -0.316228, 0.632456\}$

E=4.63268 {−0.468641, −0.320392, 0.596196, −0.320392, −0.468641}

E=2. $\{-0.316228, -0.632456, -3.4399\times10^{-16}, 0.632456, 0.316228\}$

```
E=-1.2473    {0.13114, 0.475207, 0.716911, 0.475207, 0.13114}
```

U=4:

```
E=24.3445    {0.696393, -0.119961, 0.0359561, -0.119961, 0.696393}
E=24.3246    {-0.697976, 0.113266, -9.24073×10^-17, -0.113266, 0.697976}
E=13.712     {0.115946, 0.596426, -0.511531, 0.596426, 0.115946}
E=11.6754    {-0.113266, -0.697976, -4.39734×10^-15, 0.697976, 0.113266}
E=5.94345    {0.0399186, 0.360396, 0.858512, 0.360396, 0.0399186}
```

In the attractive case (U < 0), the system can gain energy by attracting additional particles from the environment. With positive or vanishing chemical potential $\mu >= 0$ the ground state has infinitely many particles.
For U=1, μ=0, the ground state among those considered here has N=3 particles. The first excitations have 2 and 4 particles, respectively. Therefore a higher cut-off than N=4 may make sense.
For U=4, μ=0, the ground state has N=1 one particle and all states with more than 2 particles have such high energies that they are probably irrelevant for most purposes.

Fermi - Hubbard model

$$H_{FH} = -t \sum_{\langle i,j \rangle} \left(c_i^\dagger c_j + c_i c_j^\dagger \right) + \frac{U}{2} \sum_i n_i (n_i - 1) - \mu \sum_i n_i$$

Basis

Basis representation : Number of up and down spins on site 1, number of up and down spins on site 2.
We are dealing with fermions, so each site can old at most one particle of each spin.

```
FHBasis = Flatten[ Table[ {i, j, k, l}, {i, 0, 1}, {j, 0, 1}, {k, 0, 1}, {l, 0, 1}], 3]
```

```
{{0, 0, 0, 0}, {0, 0, 0, 1}, {0, 0, 1, 0}, {0, 0, 1, 1}, {0, 1, 0, 0},
 {0, 1, 0, 1}, {0, 1, 1, 0}, {0, 1, 1, 1}, {1, 0, 0, 0}, {1, 0, 0, 1},
 {1, 0, 1, 0}, {1, 0, 1, 1}, {1, 1, 0, 0}, {1, 1, 0, 1}, {1, 1, 1, 0}, {1, 1, 1, 1}}
```

Hamiltonian

```
HFH[{kf_, lf_, mf_, nf_}, {ki_, li_, mi_, ni_}] :=
 -t (DiscreteDelta[kf - ki - 1, mf - mi + 1, lf - li, nf - ni] +
     DiscreteDelta[kf - ki + 1, mf - mi - 1, lf - li, nf - ni] + DiscreteDelta[kf - ki, mf - mi,
      lf - li - 1, nf - ni + 1] + DiscreteDelta[kf - ki, mf - mi, lf - li + 1, nf - ni - 1]) +
  U (kf lf + mf nf) DiscreteDelta[kf - ki, lf - li, mf - mi, nf - ni](*+
     -μ(m+n)DiscreteDelta[k-m,l-n]*)
```

```
HFHmatrix = Table[ HFH[f, i], {f, FHBasis}, {i, FHBasis} ];
HFHmatrix // MatrixForm
```

$$\begin{pmatrix}
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & U & 0 & 0 & -t & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & U & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 \\
0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & U & 0 & 0 & -t & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & -t & 0 & 0 & U & 0 & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & U & 0 & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 & U & 0 \\
0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 2U
\end{pmatrix}$$

Sectors

N = 0

```
FHN0Basis = FHBasis[[{1}]]
```

```
{{0, 0, 0, 0}}
```

```
HFHN0matrix = Table[ HFH[f, i], {f, FHN0Basis}, {i, FHN0Basis} ];
HFHN0matrix // MatrixForm
```

$$\begin{pmatrix} 0 \end{pmatrix}$$

N = 1, $S_z = -1/2$

```
FHN1SdownBasis = FHBasis[[{2, 5}]]
```

```
{{0, 0, 0, 1}, {0, 1, 0, 0}}
```

```
HFHN1Sdownmatrix = Table[ HFH[f, i], {f, FHN1SdownBasis}, {i, FHN1SdownBasis} ];
HFHN1Sdownmatrix // MatrixForm
```

$$\begin{pmatrix} 0 & -t \\ -t & 0 \end{pmatrix}$$

```
Eigensystem[HFHN1Sdownmatrix]
```

```
{{-t, t}, {{1, 1}, {-1, 1}}}
```

N = 1, $S_z = +1/2$

```
FHN1SupBasis = FHBasis[[{3, 9}]]
```

```
{{0, 0, 1, 0}, {1, 0, 0, 0}}
```

```
HFHN1Supmatrix = Table[ HFH[f, i], {f, FHN1SupBasis}, {i, FHN1SupBasis} ];
HFHN1Supmatrix // MatrixForm
```

$$\begin{pmatrix} 0 & -t \\ -t & 0 \end{pmatrix}$$

```
Eigensystem[HFHN1Supmatrix]
```

```
{{-t, t}, {{1, 1}, {-1, 1}}}
```

N = 2, $S_z = -1$

```
FHN2SdownBasis = FHBasis[[{6}]]
```

```
{{0, 1, 0, 1}}
```

```
HFHN2Sdownmatrix = Table[ HFH[f, i], {f, FHN2SdownBasis}, {i, FHN2SdownBasis} ];
HFHN2Sdownmatrix // MatrixForm
```

$$\begin{pmatrix} 0 \end{pmatrix}$$

N = 2, $S_z = 0$

```
FHN2S0Basis = FHBasis[[{4, 7, 10, 13}]]
```

```
{{0, 0, 1, 1}, {0, 1, 1, 0}, {1, 0, 0, 1}, {1, 1, 0, 0}}
```

```
HFHN2S0matrix = Table[ HFH[f, i], {f, FHN2S0Basis}, {i, FHN2S0Basis} ];
HFHN2S0matrix // MatrixForm
```

$$\begin{pmatrix} U & -t & -t & 0 \\ -t & 0 & 0 & -t \\ -t & 0 & 0 & -t \\ 0 & -t & -t & U \end{pmatrix}$$

```
Eigensystem[HFHN2S0matrix]
```

$$\left\{\left\{0, U, \frac{1}{2}\left(U-\sqrt{16 t^2+U^2}\right), \frac{1}{2}\left(U+\sqrt{16 t^2+U^2}\right)\right\}, \left\{\{0, -1, 1, 0\}, \{-1, 0, 0, 1\}, \left\{1, -\frac{-U-\sqrt{16 t^2+U^2}}{4 t}, -\frac{-U-\sqrt{16 t^2+U^2}}{4 t}, 1\right\}, \left\{1, -\frac{-U+\sqrt{16 t^2+U^2}}{4 t}, -\frac{-U+\sqrt{16 t^2+U^2}}{4 t}, 1\right\}\right\}\right\}$$

```
N[% /. {t → 1, U → 1}]
```

```
{{0., 1., -1.56155, 2.56155}, {{0., -1., 1., 0.}, {-1., 0., 0., 1.},
  {1., 1.28078, 1.28078, 1.}, {1., -0.780776, -0.780776, 1.}}}
```

- **N = 2, $S_z = +1$**

```
FHN2SupBasis = FHBasis[[{11}]]
```

```
{{1, 0, 1, 0}}
```

```
HFHN2Supmatrix = Table[ HFH[f, i], {f, FHN2SupBasis}, {i, FHN2SupBasis} ];
HFHN2Supmatrix // MatrixForm
```

$$\begin{pmatrix} 0 \end{pmatrix}$$

- **N = 3, $S_z = -1/2$**

```
FHN3SdownBasis = FHBasis[[{8, 14}]]
```

```
{{0, 1, 1, 1}, {1, 1, 0, 1}}
```

```
HFHN3Sdownmatrix = Table[ HFH[f, i], {f, FHN3SdownBasis}, {i, FHN3SdownBasis} ];
HFHN3Sdownmatrix // MatrixForm
```

$$\begin{pmatrix} U & -t \\ -t & U \end{pmatrix}$$

```
Eigensystem[HFHN3Sdownmatrix]
```

```
{{-t + U, t + U}, {{1, 1}, {-1, 1}}}
```

- **N = 3, $S_z = +1/2$**

```
FHN3SupBasis = FHBasis[[{12, 15}]]
```

```
{{1, 0, 1, 1}, {1, 1, 1, 0}}
```

```
HFHN3Supmatrix = Table[ HFH[f, i], {f, FHN3SupBasis}, {i, FHN3SupBasis} ];
HFHN3Supmatrix // MatrixForm
```

$$\begin{pmatrix} U & -t \\ -t & U \end{pmatrix}$$

```
Eigensystem[HFHN3Supmatrix]
```

```
{{-t + U, t + U}, {{1, 1}, {-1, 1}}}
```

- **N = 4, $S_z = 0$**

```
FHN4S0Basis = FHBasis[[{16}]]
```

```
{{1, 1, 1, 1}}
```

```
HFHN4S0matrix = Table[ HFH[f, i], {f, FHN4S0Basis}, {i, FHN4S0Basis} ];
HFHN4S0matrix // MatrixForm
```

$$\begin{pmatrix} 2U \end{pmatrix}$$

t - J model

$$H_{tJ} = -t \sum_{<i,j>,\sigma} \left(c^{\dagger}_{i,\sigma} c_{j,\sigma} + c_{i,\sigma} c^{\dagger}_{j,\sigma}\right) + J \sum_{<i,j>} \left(S_i . S_j - n_i n_j / 4\right)$$

Basis

For each basis state, the first to elements are either a spinor for the particle on site 1 or 0 for no particle and the last two elements likewise for the second site.

```
tJBasis = Flatten[Table[{k, l, m, n}, {k, 0, 1}, {l, 0, 1 - k}, {m, 0, 1}, {n, 0, 1 - m}], 3]
```

```
{{0, 0, 0, 0}, {0, 0, 0, 1}, {0, 0, 1, 0}, {0, 1, 0, 0},
 {0, 1, 0, 1}, {0, 1, 1, 0}, {1, 0, 0, 0}, {1, 0, 0, 1}, {1, 0, 1, 0}}
```

Hamiltonian

```
HtJ[{kf_, lf_, mf_, nf_}, {ki_, li_, mi_, ni_}] :=
 -t (DiscreteDelta[kf - ki - 1, mf - mi + 1, lf - li, nf - ni] +
     DiscreteDelta[kf - ki + 1, mf - mi - 1, lf - li, nf - ni] +
     DiscreteDelta[kf - ki, mf - mi, lf - li - 1, nf - ni + 1] +
     DiscreteDelta[kf - ki, mf - mi, lf - li + 1, nf - ni - 1]) +
  J/4 (Sum[{kf, lf}.PauliMatrix[α].{ki, li} * {mf, nf}.PauliMatrix[α].{mi, ni}, {α, 3}] -
     (ki + li) (mi + ni) DiscreteDelta[kf - ki, mf - mi, lf - li, nf - ni])
```

```
HtJmatrix = Table[HtJ[f, i], {f, tJBasis}, {i, tJBasis}];
HtJmatrix // MatrixForm
```

$$\begin{pmatrix} 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & -t & 0 & 0 \\ 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & -\frac{J}{2} & 0 & \frac{J}{2} & 0 \\ 0 & 0 & -t & 0 & 0 & 0 & 0 & 0 & 0 \\ 0 & 0 & 0 & 0 & 0 & \frac{J}{2} & 0 & -\frac{J}{2} & 0 \\ 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 & 0 \end{pmatrix}$$

Sectors

N = 0, S_z = 0

```
tJN0Basis = tJBasis[[{1}]]
```

```
{{0, 0, 0, 0}}
```

```
HtJN0 = Table[HtJ[f, i], {f, tJN0Basis}, {i, tJN0Basis}];
HtJN0 // MatrixForm
```

$$\begin{pmatrix} 0 \end{pmatrix}$$

N = 1, S_z = -1 / 2

```
tJN1SdownBasis = tJBasis[[{2, 4}]]
```

```
{{0, 0, 0, 1}, {0, 1, 0, 0}}
```

```
HtJN1Sdown = Table[HtJ[f, i], {f, tJN1SdownBasis}, {i, tJN1SdownBasis}];
HtJN1Sdown // MatrixForm
```

$$\begin{pmatrix} 0 & -t \\ -t & 0 \end{pmatrix}$$

```
Eigensystem[HtJN1Sdown]
```

```
{{-t, t}, {{1, 1}, {-1, 1}}}
```

N = 1, S_z = +1 / 2

```
tJN1SupBasis = tJBasis[[{3, 7}]]
```

```
{{0, 0, 1, 0}, {1, 0, 0, 0}}
```

```
HtJN1Sup = Table[ HtJ[f, i], {f, tJN1SupBasis}, {i, tJN1SupBasis} ];
HtJN1Sup // MatrixForm
```

$$\begin{pmatrix} 0 & -t \\ -t & 0 \end{pmatrix}$$

```
Eigensystem[HtJN1Sup]
```

{{-t, t}, {{1, 1}, {-1, 1}}}

- **N = 2, S_z = 0**

```
tJN2S0Basis = tJBasis[[{6, 8}]]
```

{{0, 1, 1, 0}, {1, 0, 0, 1}}

```
HtJN2S0 = Table[ HtJ[f, i], {f, tJN2S0Basis}, {i, tJN2S0Basis} ];
HtJN2S0 // MatrixForm
```

$$\begin{pmatrix} -\frac{J}{2} & \frac{J}{2} \\ \frac{J}{2} & -\frac{J}{2} \end{pmatrix}$$

```
Eigensystem[HtJN2S0]
```

{{0, -J}, {{1, 1}, {-1, 1}}}

- **N = 2, S_z = -1**

```
tJN2SdownBasis = tJBasis[[{5}]]
```

{{0, 1, 0, 1}}

```
HtJN2Sdown = Table[ HtJ[f, i], {f, tJN2SdownBasis}, {i, tJN2SdownBasis} ];
HtJN2Sdown // MatrixForm
```

(0)

- **N = 2, S_z = +1**

```
tJN2SupBasis = tJBasis[[{9}]]
```

{{1, 0, 1, 0}}

```
HtJN2Sup = Table[ HtJ[f, i], {f, tJN2SupBasis}, {i, tJN2SupBasis} ];
HtJN2Sup // MatrixForm
```

(0)